
1. La lección perdida de Feynman

Parafraseo de la lección perdida de Feynman (Feynman’s lost lecture) y
de la presentación de la misma en 3B1B.1

1La lección perdida de Feynman © 2024 by Luis Mochán está licenciada mediante CC
BY-SA 4.0.

1.1. Geometría

Considere un círculo. Considere un punto excéntrico. Trace líneas desde
dicho punto al círculo (fig. 1)

Figura 1: Círculo y líneas desde un punto excéntrico.

Ahora rote todas estas líneas un ángulo recto alrededor de su punto medio
(fig. 2).

La envolvente de estas líneas forma una figura que resulta ser una elipse.
Para demostrarlo, consideremos una línea que vaya del centro del círculo a
un punto en su periferia (fig. 3).

Limpiamos la imágen removiendo todas las líneas que sobran y rotulando
los puntos relevantes (fig. 4).

Aquí, S es el punto excéntrico elegido, P es un punto cualquiera sobre el
círculo centrado en S′ y de radio r, UV es la mediatriz del segmento SP , i.e.,

Figura 2: Líneas de la fig. 1 rotadas un ángulo recto alrededor de su punto
medio

Figura 3: Figura 2 con una línea adicional desde el centro.

4



Figura 4: Algunas líneas de la fig. f:radio.

ortogonal y que pasa por su punto medio Q, R es la intersección del segmento
S′P con la línea UV y T es un punto cualquiera sobre dicha línea. Como
UV es ortogonal a PS, entonces RS = RP . Luego, S′R + RS = S′R + RP
y como S′, R y P están sobre una línea recta, S′R + RS = S′P = r, el
radio del círculo, constante, independiente del punto P que hayamos elegido.
Esto implica que el lugar geométrico de todos los puntos R obtenidos de esta
manera (para todos los posibles puntos P sobre el círculo) forman una elipse
con focos en S y S. Por otro lado ST + SP = ST + TS > r para cualquier
punto T sobre la recta que difiera de R, y por lo tanto, son puntos fuera de
la elipse. Esto muestra que la envolvente de todas las líneas UV construidas
de esta manera forman una elipse con focos en S′ y S y con semieje mayor
r/2.

1.2. Ímpetu angular

Considere ahora un cuerpo de masa m moviendose de acuerdo a la ley
de gravitación universal F = GMmr̂/r2, con G la constante de gravitación
universal, M la masa del cuerpo central (el Sol) y r la posición de la partícula.
Como la fuerza es central, el movimiento es tal que se cumple la conservación

del ímpetu angular, L = r×p, i.e., L̇ = d
dt(r×p) = ṙ×p+r× ṗ. El primer

término es cero pues la velocidad v = ṙ es paralela al ímpetu p = mv.
El segundo término es cero pues la fuerza F = ṗ es central, i.e., paralela
a r. Luego, L̇ = 0 y L es constante. Una interpretación geométrica es la
segunda ley de Kepler, que establece que áreas barridas por el radio vector
en tiempos iguales son iguales, dA/dt es constante. El área barrida en un
tiempo ∆t pequeño es ∆A = rh/2 = (r∆r sinϕ)/2 = ||r× v||∆t/2 (fig. 5) y
es proporcional a ∆t pues r × v es constante. Por otro lado, h ≈ r∆θ, por
lo cual ∆A ≈ r2∆θ/2.

Figura 5: Área barrida por el vector r en un tiempo ∆t.

1.3. Velocidad

Consideremos ahora la velocidad (fig. 6).
Durante el tiempo ∆t en que la posición cambia por un ángulo ∆θ el

área barrida es ∆A = r(t)2∆θ ∝ ∆t. Durante dicho tiempo la velocidad
cambia por ∆v = a∆t donde a = F /m = −GM r̂/r2 es la aceleración de la
gravedad. La dirección de ∆v es opuesta a la dirección de r, mientras que su
tamaño es ||∆v|| ∝ ∆t/r2 ∝ ∆θ. Por lo tanto, la punta del vector velocidad
forma un polígono (fig. 7).

Como todas las ∆v son del mismo tamaño y están todas rotadas por el
mismo ángulo una detrás de otra, poniendo una detrás de otra formamos un
círculo en el límite de ∆θ’s pequeñas. Como la punta del vector velocidad

Figura 6: Cambio de velocidad ∆v cuando la posición cambia por un ángulo
∆θ.

v(t+∆t) = v(t) +∆v(t), las puntas de los vectores velocidad describen un
círculo conforme transcurre el tiempo.

1.4. Órbita

Notamos que los vectores ∆v son antiparalelos a los vectores de posición
r. Por lo tanto, si roto el diagrama 7 un ángulo recto en la dirección de las
manecillas del reloj (fig. 8) y luego rotamos todos los vectores ∆v alrededor
de su base, apuntarían al centro del círculo (en el límite (∆θ → 0). Si ade-
más rotamos cada vector velocidad v alrededor de su centro, su envolvente
formaría una elipse (fig. 9).

En esta figura, las flechas azules coinciden con las de ∆v, que son opuestas
a la dirección del vector de posición r. Las flechas negras tienen la dirección
de la velocidad y debe ser por tanto tangente a la trayectoria del cuerpo.
Vemos en la figura que para cada dirección, la velocidad es tangencial a la
elipse que hemos construido, demostrando así que el movimiento de cuerpos
sujetos a la fuerza gravitacional de un cuerpo central, la cual decae como
1/r2, tiene la forma de una elipse, i.e., la elipse es la curva integral del campo
de velocidades.

Figura 7: Conforme cambia la dirección de un cuerpo respecto al sol por un
ángulo ∆θ, los cambios en la velocidad conservando su tamaño ∆v ∝ ∆θ

8



Figura 8: Figura 7 rotada 90 grados en la dirección de las manecillas del reloj
(con todo y letreros).

Figura 9: Figura 8 con cada vector velocidad rotado de regreso alrededor de
su centro.

10



1.5. Código metapost

1.5.1. Encabezado

g_use_svg = 1; %0 for postscript, 1 for svg
prologues:=3;
u=1cm;
outputtemplate := "%j-%c.eps";
if g_use_svg > 0:

outputtemplate := "%j-%c.svg";
outputformat := "svg";

fi

1.5.2. Muestra círculo, foco y rayos

Rotando 90 grados las líneas que llevan de puntos de un círculo al borde
se genera una elipse. Lo hago gradualmente:

pair focus;
focus=(3u,0);
beginfig(1)

save p;
path p[];
p0=fullcircle scaled (2*5u) shifted -focus;
draw p0 withpen pencircle scaled 2pt withcolor blue;
for i=0 upto 35:

draw (0,0)--((0,0)--10u*dir(i/36*360)) intersectionpoint p0;
endfor

endfig;

1.5.3. Muestra rayos rotados

beginfig(2)
save p;
path p[];
pair r[];
p0=fullcircle scaled (2*5u) shifted -focus;
draw p0 withpen pencircle scaled 2pt withcolor blue;
for i=0 upto 35:

r0:=((0,0)--10u*dir(i/36*360)) intersectionpoint p0;
p1:=(0,0)--r0;
p2:=(-unitvector(r0)--unitvector(r0))scaled 5u rotated 90 shifted (.5r0);

draw p1;
draw p2 withcolor red;
draw p1 intersectionpoint p2 withpen pencircle scaled 4pt;

endfor
draw -focus withcolor .5green withpen pencircle scaled 4pt;

endfig;

1.5.4. Muestra un rayo desde el centro del círculo

beginfig(3)
save p;
path p[];
pair r[];
p0=fullcircle scaled (2*5u) shifted -focus;
draw p0 withpen pencircle scaled 2pt withcolor blue;
for i=0 upto 35:

r0:=((0,0)--10u*dir(i/36*360)) intersectionpoint p0;
p1:=(0,0)--r0;
p2:=(-unitvector(r0)--unitvector(r0))scaled 5u rotated 90 shifted (.5r0);
draw p1;
draw p2 withcolor red;
draw p1 intersectionpoint p2 withpen pencircle scaled 4pt;

endfor
draw -focus withcolor .5green withpen pencircle scaled 4pt;
draw -focus--((0,0)--10u*dir(7/36*360)) intersectionpoint p0

withcolor -5green withpen pencircle scaled 2pt;
endfig;

1.5.5. Quita todo lo demás

beginfig(4)
save p;
path p[];
pair r[];
p0=fullcircle scaled (2*5u) shifted -focus;
draw p0 withpen pencircle scaled 2pt withcolor blue;
p3:=fullcircle xscaled 5u yscaled (sqrt(5**2-(xpart(focus)/u)**2)*u) shifted -.5focus;
draw p3;
r0=((0,0)--10u*dir(7/36*360)) intersectionpoint p0;
p1=(0,0)--r0;

12



p2=(-unitvector(r0)--unitvector(r0))scaled 5u rotated 90 shifted (.5r0);
draw p1;
draw p2 withcolor red;
z0=p1 intersectionpoint p2;
draw z0 withpen pencircle scaled 4pt;
draw -focus withcolor .5green withpen pencircle scaled 4pt;
draw -focus--((0,0)--10u*dir(7/36*360)) intersectionpoint p0

withcolor -5green withpen pencircle scaled 2pt;
label.bot(btex $S$ etex, (0,0));
label.bot(btex $S’$ etex, -focus);
label.urt(btex $P$ etex, r0);
label.ulft(btex $Q$ etex, z0);
z1=(-focus--r0) intersectionpoint p3;
label.top(btex $R$ etex, z1);
z2=point .8 of p2;
draw -focus--z2;
draw z2--r0;
label.top(btex $T$ etex, z2);
draw z1--(0,0);
draw z2--(0,0);
label.lft(btex $U$ etex, point 1 of p2);
label.rt(btex $V$ etex, point 0 of p2);

endfig;

1.5.6. Segunda ley de Kepler

beginfig(5)
save p;
path p[];
z0=(0,0);
z1=10u*right rotated 30;
z2=9u*right rotated 40;
z3=((z0--10u*right) rotated -60 shifted z2) intersectionpoint (z0--z1);
fill z0--z1--z2--cycle withcolor .3blue+.7white;
draw z2--z3 dashed evenly;
drawarrow z0--z1 withpen pencircle scaled 2pt;
drawarrow z1--z2 withpen pencircle scaled 2pt;
p1=reverse(fullcircle scaled 1.5u shifted z1 cutbefore (z1--z2) cutafter (z0--z1));
draw p1;
label.lrt(btex $r$ etex, z1);

label.urt(btex $\Delta r=v\Delta t$ etex, .5[z1,z2]);
label.llft(btex $h$ etex, .5[z2,z3]);
p0= fullcircle scaled 6u cutbefore (z0--z1) cutafter (z0--z2);
drawarrow p0;
label.urt(btex $\Delta\theta$ etex, point .5 of p0);
label.urt(btex $\Delta A$ etex, .5[z1,z2] scaled .5);
label.lft(btex $\phi$ etex, point 1 of p1);

endfig;

1.5.7. Cambio de velocidad

beginfig(6)
save p;
path p[];
z0=(0,0); %origen
z1=5u*right rotated 40; %posición
z2=2u*right rotated (40+80); %velocidad
z3=z2 shifted z1; % nueva posición
z4=2.5u*right rotated(40+80+20); %nueva velocidad
drawarrow z0--z1 withpen pencircle scaled 2pt; %posición
drawarrow (z0--z2) shifted z1 withcolor blue withpen pencircle scaled 2pt; %velocidad
drawarrow z0--z3 withpen pencircle scaled 1pt; % nueva posición
drawarrow (z0--z4) shifted z3 withcolor blue withpen pencircle scaled 1pt; %nueva velocidad
p1=fullcircle scaled 5u cutbefore (z0--z1) cutafter (z0--z3);
drawarrow p1;
z5=(10u,3u);
drawarrow (z0--z2) scaled 2 shifted z5 withcolor blue withpen pencircle scaled 2pt; %velocidad
drawarrow (z0--z4) scaled 2 shifted z5 withcolor blue withpen pencircle scaled 1pt; %nueva velocidad
drawarrow (z2--z4) scaled 2 shifted z5 withcolor blue; %diferencia
label.rt(btex $r(t)$ etex, z1);
label.lft(btex $r(t+\Delta t)$ etex, z3);
label.urt(btex $v(t)\Delta t$ etex, .5z2 shifted z1);
label.urt(btex $v(t+\Delta t)\Delta t$ etex, .5z4 shifted z3);
label.top(btex $\Delta\theta$ etex, point 1 of p1);
label.rt(btex $v(t)$ etex, z2 scaled 2 shifted z5);
label.llft(btex $v(t+\Delta t)$ etex, z4 scaled 2 shifted z5);
label.top(btex $\Delta v$ etex, .5[z2,z4] scaled 2 shifted z5);

endfig;

1.5.8. Polígono de velocidades

picture figsiete;
beginfig(7)

save p, r, N;
path p[];
pair r[];
p0=fullcircle scaled 5u;
z0=(0,-1.5u);
N=16;
for i=0 upto N-1:

drawarrow z0--point i*8/N of p0;
drawarrow subpath(.1,.9) of (point i*8/N of p0--point (i+1)*8/N of p0)

withcolor blue;
endfor
drawarrow z0--point 2 of p0 withpen pencircle scaled 2pt;
z1=point 2 of p0;
z2=point (N/4+1)*8/N of p0;
z3=point (N/4+2)*8/N of p0;
z4=(z2-z1) scaled 4 shifted z1;
z5=(z3-z2) scaled 3 shifted z3;
draw z1..z4;
draw z3..z5;
drawarrow subpath(.1,.9) of (z1--z2) withpen pencircle scaled 2pt withcolor blue;
label.rt(btex $v(t)$ etex, .7[z0,point 2 of p0]);
label.top(btex $\Delta v(t)$ etex, .5[point 2 of p0, point (N/4+1)*8/N of p0]);
p1=fullcircle scaled 4u shifted z0 cutbefore (z0--z1)

cutafter (z0--z2); draw p1;
p2=fullcircle scaled 3u shifted z2 cutbefore (z2--z4)

cutafter (z2--z5);
draw p2;
label.top(btex $\Delta\theta$ etex, point .5 of p1);
label.lft(btex $\Delta\theta$ etex, point .5 of p2);
figsiete=currentpicture;

endfig;

1.5.9. Polígono rotado

beginfig(8)
currentpicture:=figsiete rotated -90;

endfig;

1.5.10. Polígono rotado y desrotado

beginfig(9)
save p, r, d;
path p[];
pair r[];
p0=fullcircle scaled 5u;
draw p0;
z0=(-1.5u,0);
N=16;
d=abs(point 1*8/N of p0-point 0 of p0);
for i=0 upto N-1:

r0:=point i*8/N of p0 rotated -90;
drawarrow (z0--r0) shifted -.5[z0,r0] rotated 90 shifted .5[z0,r0];
draw subpath(0,.9) of (point i*8/N of p0--(0,0)) withcolor blue;
drawarrow r0-- -d*unitvector(r0) shifted r0

withcolor blue;
endfor
draw fullcircle xscaled 2.5u yscaled (sqrt(2.5**2-(xpart z0/u)**2)*u) shifted .5z0;

endfig;

1.5.11. Final

end.

mpost feynman

